*=====All_Windowing_Functions_In_One_Plot==========

HOW TO PLOT THE RESULTS OF SEVERAL SEPARATE SIMULATIONS ON A SINGLE PLOT.



To plot several waveforms on the same plot, one has to call on the "unknown" plot. The setplot new statement creates this plot. In this case eight arrays/plots are assigned to it.

```
==========================================================================
*V_SIN#    NODE_P  NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
Vsig       OUT     0       DC     0      SIN(   0      1      14.3k       )

.control
setplot              new
let                  "rect"     =     0*vector(100)
let                  "hann"     =     0*vector(100)
let                  "black"    =     0*vector(100)
let                  "hamm"     =     0*vector(100)
let                  "gauss"    =     0*vector(100)
let                  "cos"      =     0*vector(100)
let                  "bart"     =     0*vector(100)
let                  "tri"      =     0*vector(100)
==========================================================================
```

One can then run several transient simulations, and then assign the results to the created arrays/plots.

```
==========================================================================
tran       .1u       1m      0       .1u
set        specwindow=      "rectangular"
spec       1k        100k    1k      v(out)
let        unknown.rect =    dB(mag(v(out)))

tran       .1u       1m      0       .1u
set        specwindow=      "hanning"
spec       1k        100k    1k      v(out)
let        unknown.hann =    dB(mag(v(out)))


==========================================================================
```

One plot statement will plot all the results.

==========

```
plot      unknown.rect   unknown.hann  unknown.black unknown.hamm unknown.gauss   unknown.cos  unknown.bart
unknown.tri vs frequency xlog ylimit -120 20
```



==========Full_Netlist_For_Copy_Paste==========

```
OnePlot_ALL_FFT_Windows
.Option srcsteps = 1 set Gmin = 1.0000E-02
*=======Circuit_Netlist=========================
*V_SIN#    NODE_P  NODE_N DC      VALUE  SIN(    V_DC    AC_MAG FREQ    DELAY   FDamp)
Vsig       OUT     0      DC      0      SIN(    0       1      14.3k         )

.control


setplot              new
let                  "rect"     =      0*vector(100)
let                  "hann"     =      0*vector(100)
let                  "black"    =      0*vector(100)
let                  "hamm"     =      0*vector(100)
let                  "gauss"    =      0*vector(100)
let                  "cos"      =      0*vector(100)
let                  "bart"     =      0*vector(100)
let                  "tri"      =      0*vector(100)

tran       .1u       1m      0       .1u
set        specwindow=      "rectangular"
spec       1k        100k    1k      v(out)

let        unknown.rect =    dB(mag(v(out)))

tran       .1u       1m      0       .1u
set        specwindow=      "hanning"
spec       1k        100k    1k      v(out)
let        unknown.hann =    dB(mag(v(out)))

tran       .1u       1m      0       .1u
set        specwindow=      "blackman"
spec       1k        100k    1k      v(out)

let        unknown.black =   dB(mag(v(out)))
```

```
tran      .1u       1m     0      .1u
set       specwindow=     "hamming"
spec      1k        100k   1k     v(out)

let       unknown.hamm =   dB(mag(v(out)))

tran      .1u       1m     0      .1u
set       specwindow=     "gaussian"
spec      1k        100k   1k     v(out)

let       unknown.gauss =  dB(mag(v(out)))

tran      .1u       1m     0      .1u
set       specwindow=     "cosine"
spec      1k        100k   1k     v(out)
let       unknown.cos =    dB(mag(v(out)))

tran      .1u       1m     0      .1u
set       specwindow=     "bartlet"
spec      1k        100k   1k     v(out)

let       unknown.bart =   dB(mag(v(out)))

tran      .1u       1m     0      .1u
set       specwindow=     "triangle"
spec      1k        100k   1k     v(out)

let       unknown.tri =    dB(mag(v(out)))

plot      unknown.rect   unknown.hann  unknown.black unknown.hamm unknown.gauss   unknown.cos  unknown.bart
unknown.tri vs frequency xlog ylimit -120 20

*destroy

.endc
.end
```

7.29.10_12.02PM
dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/